<commit_before>fun pageA () = return <xml>
  <head>
    <title>A</title>
  </head>
  <body>
    <form>
      <table>
        <tr>
          <td>Forename:</td>
          <td><textbox{#Forename}/></td>
        </tr>
        <tr>
          <td>Enter a Surname?</td>
          <td><checkbox{#EnterSurname}/></td>
        </tr>
      </table>
      <submit action={fromA} />
    </form>
  </body>
</xml>

and fromA r =
    let
        val forename = r.Forename

        fun pageB () = return <xml>
          <head>
            <title>B</title>
          </head>
          <body>
            <form>
              Surname:
              <textbox{#Surname}/>
              <submit action={pageC'} />
            </form>
            <a link={pageA ()}>Previous</a>
          </body>
        </xml>

        and pageC' r = pageC (Some r.Surname)

        and pageC surname = return <xml>
          <head>
            <title>C</title>
          </head>
          <body>
            <p>Hello {[forename]}{case surname of
                                      None => <xml/>
                                    | Some s => <xml> {[s]}</xml>}</p>
            {case surname of
                 None => <xml><a link={pageA ()}>Previous</a></xml>
               | Some _ => <xml><a link={pageB ()}>Previous</a></xml>}
          </body>
        </xml>
    in
        if r.EnterSurname then
            pageB ()
        else
            pageC None
    end

val main = pageA
<commit_msg>Slight change to venerable Nested demo, to get proper URL resolution<commit_after>fun pageA () = return <xml>
  <head>
    <title>A</title>
  </head>
  <body>
    <form>
      <table>
        <tr>
          <td>Forename:</td>
          <td><textbox{#Forename}/></td>
        </tr>
        <tr>
          <td>Enter a Surname?</td>
          <td><checkbox{#EnterSurname}/></td>
        </tr>
      </table>
      <submit action={fromA} />
    </form>
  </body>
</xml>

and fromA r =
    let
        val forename = r.Forename

        fun pageB () = return <xml>
          <head>
            <title>B</title>
          </head>
          <body>
            <form>
              Surname:
              <textbox{#Surname}/>
              <submit action={pageC'} />
            </form>
            <a link={pageA ()}>Previous</a>
          </body>
        </xml>

        and pageC' r = pageC (Some r.Surname)

        and pageC surname = return <xml>
          <head>
            <title>C</title>
          </head>
          <body>
            <p>Hello {[forename]}{case surname of
                                      None => <xml/>
                                    | Some s => <xml> {[s]}</xml>}</p>
            {case surname of
                 None => <xml><a link={pageA ()}>Previous</a></xml>
               | Some _ => <xml><a link={pageB ()}>Previous</a></xml>}
          </body>
        </xml>
    in
        if r.EnterSurname then
            pageB ()
        else
            pageC None
    end

fun main () = pageA ()

{'fraction_non_alphanumeric': 0.15926544240400667, 'fraction_numerical': 0.0, 'mean_word_length': 1.1663051337671728, 'pattern_counts': {'":': 0, '<': 127, '<?xml version=': 0, '>': 135, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 26}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 8, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '7340466', 'n_tokens_mistral': 966, 'n_tokens_neox': 938, 'n_words': 214}